
**** ARCHIVO DE REPLICACI�N DE FELIU RIBEIRO & URDINEZ (2016).****

******************************************************************

*(Abra el banco de datos en Stata 13 antes de ejecutar los comandos).

*TABLA 1: Resumen de variables independientes por presidente.
tabstat apoyo_leg aprobacion dnu honey_moon icc tam_coal d_ideologia nep, by(president)

*FIGURA 1: Densidad acumulada de la VD (apoyo legislativo).
drop cumul
cumul apoyo_leg, gen(cumul)
sort cumul
line  cumul apoyo_leg, scheme(sj) 

*FIGURA 2: comparaci�n ente apoyo legislativo al ejecutivo y legislativo en el tiempo.
egen apoyo_leg_anual=mean( apoyo_leg), by(ano)
egen apoyo_leg_anual_polex=mean( apoyo_leg) if polex==1, by(ano)
egen apoyo_leg_anual_poldom=mean( apoyo_leg) if polex==0, by(ano)
sort ano
twoway (connected apoyo_leg_anual_polex ano) ||(connected apoyo_leg_anual_poldom ano), scheme(sj)

*TABLA 2: COMPARACI�N DE MODELOS 1 y 2.
  **MODELO DE REGRESI�N ROBUSTA 1: apoyo legislativo a iniciativas del PE.
  *se ejecutar� sin mostrar los resultados. Deber� rodar el segundo modelo y el comando "esttab" m�s abajo.
   quietly rreg apoyo_leg i.polex##c.aprobacion i.polex##c.tam_coal i.polex##c.dnu c.dnu  icc tam_coal d_ideologia nep i.honey_moon  i.pres_code i.cod_casa i.ano if iniciativa_executivo==1,
   estimates store pe
  **MODELO DE REGRESI�N ROBUSTA 2: apoyo legislativo a iniciativas del PL.
   quietly rreg apoyo_leg i.polex##c.tam_coal  icc tam_coal d_ideologia nep  i.pres_code i.cod_casa i.ano if iniciativa_executivo==0,
   estimates store pl
esttab pe pl

*FIGURA 3: Efecto marginal de la aprobaci�n popular.
quietly rreg apoyo_leg i.polex##c.aprobacion i.polex##c.tam_coal i.polex##c.dnu c.dnu  icc tam_coal d_ideologia nep i.honey_moon  i.pres_code i.cod_casa i.ano if iniciativa_executivo==1,
margins  polex, at(aprobacion=(0(5)100))
marginsplot, scheme(sj)

*FIGURA 4: Efecto marginal del tama�o de la coalici�n de gobierno.
quietly rreg apoyo_leg i.polex##c.aprobacion i.polex##c.tam_coal i.polex##c.dnu c.dnu  icc tam_coal d_ideologia nep i.honey_moon  i.pres_code i.cod_casa i.ano if iniciativa_executivo==1,
margins  polex, at(tam_coal=(20(5)70))
marginsplot, scheme(sj)

*FIGURA 5: Efecto marginal del uso de DNUs(necesita rodar MODELO 1 anteriormente).
quietly rreg apoyo_leg i.polex##c.aprobacion i.polex##c.tam_coal i.polex##c.dnu dnu  icc tam_coal d_ideologia nep i.honey_moon  i.pres_code i.cod_casa i.ano if iniciativa_executivo==1,
margins  i.polex, at(dnu=(0(10)100))  estimtolerance(2)
marginsplot, scheme(sj)


**APENDICE ONLINE**
*FIGURA A: Box Plot de la variable dependiente.
graph hbox apoyo_leg_anual_polex apoyo_leg_anual_poldom, over(iniciativa_executivo) scheme(sj)

*TABLA A: Resumen de variables independientes por c�mara. 
tabstat apoyo_leg tam_coal d_ideologia nep polex iniciativa_executivo high_low, by(casa) s(mean sd min max n)

*TABLA B: Matriz de correlaci�n.
corr iniciativa_executivo polex aprobacion dnu  icc tam_coal d_ideologia nep honey_moon 


**********************************************
*NO REPORTADO EN EL PAPER: TESTS DE ROBUSTES**
**********************************************

*TABLA CON LAS 50 LEYES MAS AFECTADAS POR EL MODELO ROBUSTO EN SUS PESOS (WEIGHTS)
quietly rreg apoyo_leg i.polex##c.aprobacion i.polex##c.tam_coal i.polex##c.dnu dnu  icc tam_coal d_ideologia nep i.honey_moon  i.pres_code i.cod_casa i.ano if iniciativa_executivo==1, gen(wt)
predict p if e(sample)
predict r if e(sample), resid
predict hat if e(sample), hat
sort wt
list codigo_ley fecha p r h wt in 1/50

*residual versus fitted plot
scatter r p, yline(0)

* TEST DE ROBUSTES DE LOS R2 DE LOS MODELOS (NECESITA INSTALAR RREGFIT) ESTOS SON LOS R2 REPORTADOS EN EL TRABAJO.
findit rregfit
rreg apoyo_leg i.polex##c.aprobacion i.polex##c.tam_coal i.polex##c.dnu dnu  icc tam_coal d_ideologia nep i.honey_moon  i.pres_code i.cod_casa i.ano if iniciativa_executivo==1, 
rregfit
rreg apoyo_leg i.polex##c.tam_coal  icc tam_coal d_ideologia nep  i.pres_code i.cod_casa i.ano if iniciativa_executivo==0,
rregfit


* COMPARACION DE MODELO ROBUSTO CON OTRAS ESTIMACIONES CONSIDERADAS POR LOS AUTORES
*MODELO 1 (iniciativas presidenciales)

*OLS
reg apoyo_leg i.polex##c.aprobacion i.polex##c.tam_coal i.polex##c.dnu dnu  icc tam_coal d_ideologia nep i.honey_moon  i.pres_code i.cod_casa i.ano if iniciativa_executivo==1, r
estimates store ols

*TOBIT
tobit apoyo_leg i.polex##c.aprobacion i.polex##c.tam_coal i.polex##c.dnu dnu  icc tam_coal d_ideologia nep i.honey_moon  i.pres_code i.cod_casa i.ano if iniciativa_executivo==1, ul(1)
estimates store  tob

*REGRESION CUANTILICA
qreg apoyo_leg i.polex##c.aprobacion i.polex##c.tam_coal i.polex##c.dnu dnu  icc tam_coal d_ideologia nep i.honey_moon  i.pres_code i.cod_casa i.ano if iniciativa_executivo==1, vce(r)
estimates store  qre

esttab pe ols tob qre, r2

*MODELO 2 (iniciativas legislativas)
*OLS
reg apoyo_leg i.polex##c.tam_coal  icc tam_coal d_ideologia nep  i.pres_code i.cod_casa i.ano if iniciativa_executivo==0, r
estimates store ols2

*TOBIT
tobit apoyo_leg i.polex##c.tam_coal  icc tam_coal d_ideologia nep  i.pres_code i.cod_casa i.ano if iniciativa_executivo==0, ul(1)
estimates store  tob2

*REGRESION CUANTILICA
qreg apoyo_leg i.polex##c.tam_coal  icc tam_coal d_ideologia nep  i.pres_code i.cod_casa i.ano if iniciativa_executivo==0, vce(r)
estimates store  qre2

esttab pl ols2 tob2 qre2, r2












